IEOR 240 Case 1
Calgary Desk Company: Final Report
Prepared by: Aimee Chen, Christy Chen, Andersen Zhang, Oliver Wu
October 24, 2018

FORMULATION

Calgary Desk Company (CALDESCO) is a well-established desk manufacturing company. The linear program developed in this case is intended to provide a recommendation for CALDESCO's production schedule for various desk types in the month of September. First, the following assumptions are necessary for defining the problems presented in the case as well as developing the linear program:

1. The unit profit is in US dollars ($).
2. The unit time-requirement for each production line is in minutes (min)
3. The total available time for the two production 3 lines combined is 19200 minutes
4. The production volumes of desks can be measured in fractions, based on the assumption that the company can sell replacement parts. The price of replacement parts would be the proportion of the whole desk. For example, if the desk leg is ¼ of the table, it will be sold at a price ¼ of the desk.

Parameters

Below is a list of all relevant parameters of this linear program:

- M: number of lines = 3 (economy, basic, hand-crafted)
- N: number of size = 3 (student, standard, executive)
- $Profit_{mn}$, $m = 1...M, n = 1...N$: unit profits of each line and size [economy (student, standard, executive), basic(student, standard, executive), hand-crafted(student, standard, executive)]

  $Profit_{11} = 20$ economy student
  $Profit_{12} = 30$ economy standard
  $Profit_{13} = 40$ economy executive
  $Profit_{21} = 50$ basic student
  $Profit_{22} = 80$ basic standard
  $Profit_{23} = 125$ basic executive
  $Profit_{31} = 100$ hand-crafted student
  $Profit_{32} = 250$ hand-crafted standard
  $Profit_{33} = 325$ hand-crafted executive
- $Order_{mn}$, $m = 1...M, n = 1...N$: order amount of each line and size in Sept.

$Order_{11} = 750$ economy student
$Order_{12} = 1500$ economy standard
$Order_{13} = 100$ economy executive
$Order_{21} = 400$ basic student
$Order_{22} = 1500$ basic standard
$Order_{23} = 100$ basic executive
$Order_{31} = 25$ hand-crafted student
$Order_{32} = 150$ hand-crafted standard
$Order_{33} = 50$ hand-crafted executive

- $Aluminum_{mn}$, *m = 1...M, n = 1...N*: the amount of Aluminum (in square feet) required to produce each style of desk

$Aluminum_{11} = 14$ economy student
$Aluminum_{12} = 24$ economy standard
$Aluminum_{13} = 30$ economy executive
Others = 0

- $ParticleBoard_{mn}$, *m = 1...M, n = 1...N*: the amount of Particle Board (in square feet) required to produce each style of desk

$ParticleBoard_{11} = 8$ economy student
$ParticleBoard_{12} = 15$ economy standard
$ParticleBoard_{13} = 24$ economy executive
Others = 0

- $PineSheets_{mn}$, *m = 1...M, n = 1...N*: the amount of 1 1/2 - inch thick pine sheets (in square feet) required to produce each style of desk

$PineSheets_{21} = 22$ basic student
$PineSheets_{22} = 40$ basic standard
$PineSheets_{23} = 55$ basic executive
$PineSheets_{31} = 25$ hand-crafted student
$PineSheets_{32} = 45$ hand-crafted standard
$PineSheets_{33} = 60$ hand-crafted executive
Others = 0

- $PL1_{mn}$, *m = 1...M, n = 1...N*: time requirement (in minutes) per desk for production line 1.

$PL1_{11} = 1.5$ economy student
$PL1_{12} = 2$ economy standard
$PL1_{13} = 2.5$ economy executive
Others = 0

- $PL2_{mn}$, $m = 1...M$, $n = 1...N$: time requirement (in minutes) per desk for production line 2.

  $PL2_{11} = 1$ economy student
  $PL2_{12} = 1$ economy standard
  $PL2_{13} = 1$ economy executive
  $PL2_{21} = 1$ basic student
  $PL2_{22} = 1$ basic standard
  $PL2_{23} = 1$ basic executive
  Others = 0

- $PL3_{mn}$, $m = 1...M$, $n = 1...N$: time requirement (in minutes) per desk for production line 3.

  $PL3_{21} = 3$ basic student
  $PL3_{22} = 4$ basic standard
  $PL3_{23} = 5$ basic executive
  $PL3_{31} = 3$ hand-crafted student
  $PL3_{32} = 4$ hand-crafted standard
  $PL3_{33} = 5$ hand-crafted executive
  Others = 0

- $Assemble_{mn}$, $m = 1...M$, $n = 1...N$, time requirement (in minutes) per desk for Assembling/Finishing

  $Assemble_{11} = 10$ economy student
  $Assemble_{12} = 11$ economy standard
  $Assemble_{13} = 12$ economy executive
  $Assemble_{21} = 15$ basic student
  $Assemble_{22} = 18$ basic standard
  $Assemble_{23} = 20$ basic executive
  $Assemble_{31} = 20$ hand-crafted student
  $Assemble_{32} = 25$ hand-crafted standard
  $Assemble_{33} = 30$ hand-crafted executive

- $HandCraft_{mn}$, $m = 1...M$, $n = 1...N$, time requirement (in minutes) per desk for Hand-Crafting

  $HandCraft_{31} = 50$ hand-crafted student
  $HandCraft_{32} = 60$ hand-crafted standard
  $HandCraft_{33} = 70$ hand-crafted executive

- *Labor_Availabe*, = 230,400 (man-minutes), total labor availability in September
- *Aluminum_Available*, = 65,000 (SQ.FT.), September availability of aluminum (in square feet)

- *ParticleBoard_Available*, = 60,000 (SQ.FT.), September availability of Particle Board (in square feet)
- *PineSheets_Available*, = 175,000 (SQ.FT.), September availability of Pine Sheets (in square feet)
- *PL1_Available*, = 9600 (min.), Production times available on the production line 1 (in minutes)
- *PL2_Available,* = 9600 (min.), Production times available on the production line 2 (in minutes)
- *PL3_Available*, = 19200 (min), Production times available on the production line 3 (in minutes)
- $MinLineQuotas_m$, m = 1...M, Minimum quotas (of total production) for desk production of each line (economy, basic, hand-crafted)

  $MinLineQuotas_1$ = 20%

  $MinLineQuotas_2$ = 40%

  $MinLineQuotas_3$ = 10%
- $MinSizeQuotas_n$, n = 1...N, Minimum quotas (of total production) for desk production of each size (student, standard, executive)

  $MinSizeQuotas_1$ = 20%

  $MinSizeQuotas_2$ = 40%

  $MinSizeQuotas_3$ = 5%
- $MaxLineQuotas_m$, m = 1...M, Maximum quotas (of total production) for desk production of each line (economy, basic, hand-crafted)

  $MaxLineQuotas_1$ = 50%

  $MaxLineQuotas_2$ = 60%

  $MaxLineQuotas_3$ = 20%
- $MaxSizeQuotas_n$, n = 1...N, Maximum quotas (of total production) for desk production of each size (student, standard, executive)

  $MaxSizeQuotas_1$ = 35%

  $MaxSizeQuotas_2$ = 70%

  $MaxSizeQuotas_3$ = 15%

Decision Variable

The decision variables that will be used in this linear program is the production amount for each type of desks [economy (student, standard, executive), basic (student, standard, executive), hand-crafted (student, standard, executive)], represented as:

$$X_{mn} , m = 1...M, n = 1...N$$

Objective Function

The objective for this linear program is to maximize total profit with a particular production schedule, which is given by:

$$\text{Max} \sum_{m=1}^{M} \sum_{n=1}^{N} Profit_{mn} X_{mn}$$

Constraints

The linear program is subjected to the following constraints:

1. CALDESCO need to meet all outstanding orders for September

$$X_{mn} \geq Order_{mn} \text{ , } m = 1...M, n = 1...N$$

2. Minimum quotas (of total production) for desk production of each line (economy, basic, hand-crafted)

$$MinLineQuotas_m * \sum_{m=1}^{M} \sum_{n=1}^{N} X_{mn} \leq \sum_{n=1}^{N} X_{mn} \text{ , } m = 1...M$$

3. Maximum quotas (of total production) for desk production of each line (economy, basic, hand-crafted)

$$MaxLineQuotas_m * \sum_{m=1}^{M} \sum_{n=1}^{N} X_{mn} \geq \sum_{n=1}^{N} X_{mn} \text{ , } m = 1...M$$

4. Minimum quotas (of total production) for desk production of each size (student, standard, executive)

$$MinSizeQuotas_n * \sum_{m=1}^{M} \sum_{n=1}^{N} X_{mn} \leq \sum_{m=1}^{M} X_{mn} \text{ , } n = 1...N$$

5. Maximum quotas (of total production) for desk production of each size (student, standard, executive)

$$MaxSizeQuotas_n * \sum_{m=1}^{M} \sum_{n=1}^{N} X_{mn} \geq \sum_{m=1}^{M} X_{mn} \text{ , } n = 1...N$$

6. Amount of Aluminum used cannot exceed the availability of Aluminum

$$\sum_{m=1}^{M} \sum_{n=1}^{N} X_{mn} * Aluminum_{mn} \leq Aluminum\_Available$$

7. Amount of Particle Board used cannot exceed the availability of Particle Board

$$\sum_{m=1}^{M} \sum_{n=1}^{N} X_{mn} * ParticleBoard_{mn} \leq ParticleBoard\_Available$$

8. Amount of Pine Sheets used cannot exceed the availability of Pine Sheets

$$\sum_{m=1}^{M}\sum_{n=1}^{N} X_{mn} * PineSheets_{mn} \le PineSheets\_Available$$

9. Time used for Production Line 1 cannot exceed the time availability of Production Line 1

$$\sum_{m=1}^{M}\sum_{n=1}^{N} X_{mn} * PL1_{mn} \le PL1\_Available$$

10. Time used for Production Line 2 cannot exceed the time availability of Production Line 2

$$\sum_{m=1}^{M}\sum_{n=1}^{N} X_{mn} * PL2_{mn} \le PL2\_Available$$

11. Time used for Production Line 3 cannot exceed the time availability of Production Line 3

$$\sum_{m=1}^{M}\sum_{n=1}^{N} X_{mn} * PL3_{mn} \le PL3\_Available$$

12. Total man-minutes used cannot exceed the total man-minutes available

$$2* \sum_{m=1}^{M}\sum_{n=1}^{N} X_{mn} *(PL1_{mn} + PL2_{mn} + PL3_{mn}) + \sum_{m=1}^{M}\sum_{n=1}^{N} X_{mn} *(Assemble_{mn} + HandCraft_{mn})$$
$$\le Labor\_Availabe$$

13. Non-negativity for decision variables

$$X_{mn} \ge 0,\ m = 1...M,\ n = 1...N,$$

SOLUTION

Solving the linear program using AMPL, we found that CALDESCO will maximize its profit while meeting demand and internal quotas by following the schedule in Table 1. The maximum profit would be $612,113.

Table 1: September Production Schedule Recommended for CALDESCO

	Economy	**Basic**	**Hand-Crafted**
Student	750	525.537	25
Standard	1,500	1,657.5	1,069.24
Executive	100	825.403	50

From AMPL, we get Table 2, which details the amount of each resource needed for implementing the optimal solution.

Table 2: Various Resources Needed for Implementation

	Used	Unused	Total
Aluminum (SQ.FT.)	49,500	15,500	65,000
Particle Board (SQ.FT.)	30,900	29,100	60,000
Pine Sheets (SQ.FT.)	17,500	0	17,500
Labor (Minutes)	230,400	0	230,400
Production Line 1 (Minutes)	4,375	5,225	9,600
Production Line 2 (Minutes)	5,358	4,242	9,600
Production Line 3 (Minutes)	16,936	2,264	19,200

DISCUSSION

Sensitivity Analysis

1. *How much the unit profit of economy-line, student-size desk can go up or down from $20 without changing the current optimal production quantities?*

Table 3: Sensitivity Analysis of Production for Economy Student Desks

_varname	_var	_var.rc	_var.down	_var.up	_var.current
Economy Student	750	0	-1.00E+20	24	20

Optimal production amount for economy-line, student-size desk would remain unchanged if the unit profit goes up from \$20 to \$24, or goes down to infinity. We choose economy-line, student-sized desk as an example of basic variable.

All decision variables, in this case, are *basic variables*. In other words, when total profit is optimal, productions for its entire line of desks are all *non-zero*.

2. *How much the unit profit of other desks can go up or down without changing the current optimal production quantities? What does it imply in this case?*

All unit profits of desks can go up without changing the optimal quantities. The specific values can be found from the following table.

Table 4: Sensitivity Analysis of Production Volumes for All Desk Types

_varname	_var	_var.rc	_var.down	_var.up	_var.current
Economy Student	750	0	-1.00E+20	24	20
Econonmy Standard	1500	0	-1.00E+20	42	30
Economy Executive	100	0	-1.00E+20	78	40
Basic Student	526	0	4.54E+01	62	50
Basic Standard	1658	0	7.20E+01	83	80
Basic Executive	825	0	1.21E+02	171	125
Hard-crafted Student	25	0	-1.00E+20	188	100
Hard-crafted Standard	1069	0	2.47E+02	270	250
Hand-crafted Executive	50	0	-1.00E+20	329	325
TotalProduction	6503	0	0.00E+00	0	0

We obtain new optimal objective value through increasing unit profit of desks. The new optimal objective value is:

$$\sum_{m=1}^{M} \sum_{n=1}^{N} New\ Profit_{mn}\ X_{mn}$$

= 24*750+42*1500+78*100+62*526+83*1658+171*825+188*25+270*1069+329*50
= **710423**

The new optimal objective value is \$710,423, which is \$98,310 higher than the original optimal objective value.

3. *What is the maximum potential increase in profit when an extra unit of labor is available? What would be the new optimal objective value?*

Table 5: Sensitivity Analysis of Labor Availability

_conname	_con	_con.slack	_con.current	_con.down	_con.up
labor_availability	2.59735	0	230400	200443	239325

The shadow price of labor_availability is 2.597, which means that the total profit will increase at $2.597 per unit of labor. Hence, $2.597 is the maximum increase in profit that we will get when an extra unit of labor becomes available.

The upper bound of labor_availability constraint is 239,325, and lower bound is 200,443. Through calculation, we know that the allowable increase is 8,925, 3.87% higher than the current constraint. Increasing the labor availability to its upper bound (239,325), we will increase the optimal point by $23,178 (shadow price*allowable increase = $2.597*8,925). The new objective value increases from $612,113 to $635,291.

This is an example the change of b_i of binding constraints: a change of constraint would lead to a change of optimal point.

4. *What is the maximum potential increase in profit when an extra unit of aluminum is available? What would be the new optimal objective value?*

Table 6: Sensitivity Analysis of Aluminum Availability

_conname	_con	_con.slack	_con.current	_con.down	_con.up
aluminum_availability	0	15500	65000	49500	1.00E+20

The shadow price of aluminum_availability is 0, which means that the total profit would not change with the change of aluminum availability. When the aluminum changes within the range of (49500, +∞), the optimal solution and objective value would be unchanged. This is an example of non-binding constraint: a change of constraint would have no impact on the optimal objective value.

5. *Which constraints can we change to increase the total profits? Which constraints cannot?*

Table 7: Sensitivity Analysis of All Constraints

_conname	type	_con	_con.slack	_con.current	_con.down	_con.up
'=TotalProduction'	binding	2.10551	0	0	0	0
'meetorder[1,1]'	binding	-4.06234	0	0	0	0
'meetorder[1,2]'	binding	-12.0494	0	0	0	0
'meetorder[1,3]'	binding	-38.3373	0	0	0	0
'meetorder[2,1]'	non-binding	0	125.537	0	0	0
'meetorder[2,2]'	non-binding	0	157.501	0	0	0
'meetorder[2,3]'	non-binding	0	725.403	0	0	0
'meetorder[3,1]'	binding	-88.3629	0	0	0	0
'meetorder[3,2]'	non-binding	0	919.244	0	0	0
'meetorder[3,3]'	binding	-3.7655	0	0	0	0
'minline[1]'	non-binding	0	1049.46	0	-1049.46	1.00E+20
'minline[2]'	non-binding	0	407.367	0	-407.367	1.00E+20
'minline[3]'	non-binding	0	493.976	0	-493.976	1.00E+20
'maxline[1]'	non-binding	0	901.343	0	-1.00E+20	901.343
'maxline[2]'	non-binding	0	893.17	0	-1.00E+20	893.17
'maxline[3]'	binding	0	156.293	0	-1.00E+20	156.293
'minprod[1]'	binding	12.7924	2.27E-13	0	~~-281.915~~	114.494
'minprod[2]'	non-binding	0	1625.67	0	-1625.67	1.00E+20
'minprod[3]'	non-binding	0	650.269	0	-650.269	1.00E+20
'maxprod[1]'	non-binding	0	975.403	0	-1.00E+20	975.403
'maxprod[2]'	non-binding	0	325.134	0	-1.00E+20	325.134
'maxprod[3]'	binding	-31.0932	-1.14E-13	0	~~-114.905~~	331.807
aluminum_availability	binding	0	15500	65000	49500	1.00E+20
PB_availability	non-binding	0	29100	60000	30900	1.00E+20
PS_availability	binding	0.234495	0	175000	169646	195940
PL1_availability	non-binding	0	5225	0	0	0
PL2_availability	non-binding	0	4241.56	9600	5358.44	1.00E+20
PL3_availability	non-binding	0	2264.39	19200	16935.6	1.00E+20
labor_availability	binding	2.59735	0	230400	200443	239325

If we increase binding constraints up to their upper bound, we would increase the total profits by the product of shadow price times allowable increase. Labor_availability is an example. From this table, we label constraints with binding/non-binding, and binding constraints are not highlighted with light orange. In addition, we striked through the lower bounds of minprod[1] and maxprod[3], because their lower bounds are negative values and productions cannot be non-positive.

Real Life Factors to Consider

We need to consider the following real-life factors:

1. The order of production: first manufacturing and then assembling.
As each desk requires different parts (drawers, base, and top), we need to first manufacture the parts and then assembly. In other words, no workers could assemble parts before parts are produced. However, as it only takes a short time to manufacture a single part, there would always be parts available for assembly. No extra constraint is needed.

2. Total labor availability is not guaranteed for the optimal objective value.
The optimal objective value is based on the assumption that the total labor availability is 230,400. However, we need to consider the fact that 1+1 is not necessarily 2. For example, the assembly of a handcrafted student table requires 20 minutes from a single worker. If we assign this task to 2 workers, it is not guaranteed that the assembly minutes would be equally split between them. We cannot say that two workers work for 10 minutes together and then the handcrafted student table is assembled. However, the model is built on such an assumption. We need to take this into account in real life.

3. Materials may have waste, and therefore not all materials will be converted into the product.
It is assumed that all raw materials are converted into desks without material wastes. However, in real life, there are material wastes. We can include this factor in the model with conversion rates. For example, for the constraint of aluminum availability, we can multiply the left-hand side constraint by 0.8. Constraints of production line, type and size should all take this factor into account.

4. Material suppliers normally sell in bulk, so CALDESCO may not purchase materials in the optimal amount.
In real-life situations, suppliers of raw material often do not provide the exact amount of materials needed but sell them in bulk. For example, if the company needs 750 square-feet of aluminum and the supplier only sells aluminum in 1000 square-feet bulk, the company has to purchase 1000 square-feet. To include this factor, we can add constraints and we need integer linear programming.

5. It takes time for a craftsman to switch tasks.
It is assumed that all craftsman can carry out all the tasks, and labor hour is identical across all the tasks. For example, in the first hour, a craftsman controls a production line and then assembles desks in the second hour. However, we should not take for granted that the craftsman

would use 2 hours to finish these two tasks, because there is time cost to switch tasks. If we need to consider switching costs, we would need additional constraints. The number of constraints will depend on the difference in time costs between tasks. The same argument can also be made regarding monetary costs when switching tasks.

6. Profit margin decreases as sales volumes increase.

When CALDESCO sells desks in a high volume, the profit margin would decrease. However, in our model, we assume a constant profit margin for each desk type/size. This is not realistic. For a corporate client, the price may be lowered and therefore the profit margin would be lower. In our case, we can consider the profit margin as the average profit margin, and therefore we have already taken this factor into account these situations.

RECOMMENDATION

Based on our analysis thus far, we suggest CALDESCO with the following production plan in Table 1.

Table 1: September Production Schedule Recommended for CALDESCO

	Economy	**Basic**	**Hand-Crafted**
Student	750	525.537	25
Standard	1,500	1,657.5	1,069.24
Executive	100	825.403	50

During production process, however, many circumstances may change. For example, the availability of aluminum may increase or decrease. In this situation, we can increase the aluminum availability to improve profitability. For non-binding constraints, the production plan does not need to change as long as the constraints change within the range between upper bound and lower bound. For changes in binding constraints, the optimal solution needs to be re-calculated and the total profit will be affected based on shadow price.

Once the production is finished, we can further improve total profit through adjusting the unit profit of desk (objective coefficients of the primal problem). With the unit profit given in the prompt, the optimal value of total profit is $612,113. If we increase objective coefficients up to their upper bounds based on Table 4, the optimal value of total profit will increase. The optimal value can be improved to $710,423.

Appendix – AMPL files

Case1.mod

```
# Parameters
param M; # Number of lines (economy, basic, and hand-crafted)
param N; # Number of sizes (student, standard, and executive)
param Profit{m in 1..M, n in 1..N}; # Unit profit of each style of
desks
param Order{m in 1..M, n in 1..N}; # Orders for each style of desks
param Aluminum{m in 1..M, n in 1..N}; # Aluminum required to produce
each style of desks
param ParticleBoard{m in 1..M, n in 1..N}; # Particle board required
to produce each style of desks
param PineSheets{m in 1..M, n in 1..N}; # Pine sheet required to
produce each style of desks
param PL1{m in 1..M, n in 1..N}; # minutes required per desk for
production line 1
param PL2{m in 1..M, n in 1..N}; # minutes required per desk for
production line 2
param PL3{m in 1..M, n in 1..N}; # minutes required per desk for
production line 3
param Assemble{m in 1..M, n in 1..N}; # minutes required per desk for
assembling/finishing
param HandCraft{m in 1..M, n in 1..N}; # minutes required per desk for
hand-crafting
param Labor_Available; # total labor available in September
param Aluminum_Available; # total aluminum available
param ParticleBoard_Available; # total particle boards available
param PineSheets_Available; # total pine sheets available
param PL1_Available; # production minutes available on production line
1
param PL2_Available; # production minutes available on production line
2
param PL3_Available; # production minutes available on production line
3
param MinLineQuotas{m in 1..M}; # minimum quotas for desk production
of each line (econ, basic, handcrafted)
param MinSizeQuotas{n in 1..N}; # minimum quotas for desk production
of each size (student, standard, exec)
param MaxLineQuotas{m in 1..M}; # maximum quotas for desk production
of each line (econ, basic, handcrafted)
param MaxSizeQuotas{n in 1..N}; # maximum quotas for desk production
of each size (student, standard, exec)
```

```
# Decision Variable
var X{m in 1..M, n in 1..N} >= 0; # Amount to produce for each desk
type
var TotalProduction = sum{m in 1..M, n in 1..N} X[m,n];

# Objective Function
maximize totalprofit:
	sum{m in 1..M, n in 1..N} X[m,n]*Profit[m,n];

# Constraints
subject to meetorder {m in 1..M, n in 1..N}:
	X[m,n] >= Order[m,n];

subject to minline {m in 1..M}:
	MinLineQuotas[m] * TotalProduction <= sum{n in 1..N} X[m,n];

subject to maxline {m in 1..M}:
	MaxLineQuotas[m] *  TotalProduction>= sum{n in 1..N} X[m,n];

subject to minprod {n in 1..N}:
	MinSizeQuotas[n] * TotalProduction <= sum{m in 1..M} X[m,n];

subject to maxprod {n in 1..N}:
	MaxSizeQuotas[n] * TotalProduction >= sum{m in 1..M} X[m,n];

subject to aluminum_availability:
	sum{m in 1..M, n in 1..N} (X[m,n] * Aluminum[m,n]) <=
Aluminum_Available;

subject to PB_availability:
	sum{m in 1..M, n in 1..N} (X[m,n] * ParticleBoard[m,n]) <=
ParticleBoard_Available;

subject to PS_availability:
	sum{m in 1..M, n in 1..N} (X[m,n] * PineSheets[m,n]) <=
PineSheets_Available;

subject to PL1_availability:
	sum{m in 1..M, n in 1..N} (X[m,n] * PL1[m,n]) <= PL1_Available;

subject to PL2_availability:
	sum{m in 1..M, n in 1..N} (X[m,n] * PL2[m,n]) <= PL2_Available;

subject to PL3_availability:
```

```
    sum{m in 1..M, n in 1..N} (X[m,n] * PL3[m,n]) <= PL3_Available;

subject to labor_availability:
    (sum{m in 1..M, n in 1..N} (X[m,n] *
(PL1[m,n]+PL2[m,n]+PL3[m,n]))) * 2 +
    sum{m in 1..M, n in 1..N} (X[m,n] *
(Assemble[m,n]+HandCraft[m,n]))
    <= Labor_Available;
```

Case1.dat

```
param M := 3; # Number of lines (economy, basic, and hand-crafted)
param N := 3; # Number of products (student, standard, and executive)
param Profit :=
    1 1 20
    1 2 30
    1 3 40
    2 1 50
    2 2 80
    2 3 125
    3 1 100
    3 2 250
    3 3 325; # Unit profit of each style of desks
param Order :=
    1 1 750
    1 2 1500
    1 3 100
    2 1 400
    2 2 1500
    2 3 100
    3 1 25
    3 2 150
    3 3 50; # Orders for each style of desks
param Aluminum :=
    1 1 14
    1 2 24
    1 3 30
    2 1 0
    2 2 0
    2 3 0
    3 1 0
    3 2 0
    3 3 0; # Aluminum required to produce each style of desks
```

```
param ParticleBoard :=
      1 1 8
      1 2 15
      1 3 24
      2 1 0
      2 2 0
      2 3 0
      3 1 0
      3 2 0
      3 3 0; # Particle board required to produce each style of desks
param PineSheets :=
      1 1 0
      1 2 0
      1 3 0
      2 1 22
      2 2 40
      2 3 55
      3 1 25
      3 2 45
      3 3 60; # Pine sheet required to produce each style of desks
param PL1 :=
      1 1 1.5
      1 2 2
      1 3 2.5
      2 1 0
      2 2 0
      2 3 0
      3 1 0
      3 2 0
      3 3 0; # minutes required per desk for production line 1
param PL2 :=
      1 1 1
      1 2 1
      1 3 1
      2 1 1
      2 2 1
      2 3 1
      3 1 0
      3 2 0
      3 3 0; # minutes required per desk for production line 2
param PL3 :=
      1 1 0
      1 2 0
      1 3 0
```

```
      2 1 3
      2 2 4
      2 3 5
      3 1 3
      3 2 4
      3 3 5; # minutes required per desk for production line 3
param Assemble :=
      1 1 10
      1 2 11
      1 3 12
      2 1 15
      2 2 18
      2 3 20
      3 1 20
      3 2 25
      3 3 30; # minutes required per desk for assembling/finishing
param HandCraft :=
      1 1 0
      1 2 0
      1 3 0
      2 1 0
      2 2 0
      2 3 0
      3 1 50
      3 2 60
      3 3 70; # minutes required per desk for hand-crafting
param Labor_Available := 230400; # total labor available in September
param Aluminum_Available := 65000; # total aluminum available
param ParticleBoard_Available := 60000; # total particle boards
available
param PineSheets_Available := 175000; # total pine sheets available
param PL1_Available := 9600; # production minutes available on
production line 1
param PL2_Available := 9600; # production minutes available on
production line 2
param PL3_Available := 19200; # production minutes available on
production line 3
param MinLineQuotas :=
      1 0.2
      2 0.4
      3 0.1; # minimum quotas for desk production of each line (econ,
basic, handcrafted)
param MinSizeQuotas :=
      1 0.2
```

```
     2 0.4
     3 0.05; # minimum quotas for desk production of each size
(student, standard, exec)
param MaxLineQuotas :=
     1 0.5
     2 0.6
     3 0.2; # maximum quotas for desk production of each line (econ,
basic, handcrafted)
param MaxSizeQuotas :=
     1 0.35
     2 0.7
     3 0.15; # maximum quotas for desk production of each size
(student, standard, exec)
```